Deutsche Airbus

# Technical Note

Copy to: Dr. Dey EV 5
Dr. Seidel JET - E
H. Fischer EV 52

Page of pages

further distribution arranged on request

**Document No.:** TN-EV52-360/91

**Department:** EV 52

**Project:** MPC 75

**System:** Flight Controls, ATA 27

**Subject:** Equations for a preliminary actuator design

**Reference:**

**Summary:** This note summarizes equations used for a preliminary actuator design. The equations can be applied for the preliminary design of elevator, aileron, rudder, THS, and spoiler actuators.

| | Prepared: | Checked: | Approved: | Authorized: |
|---|---|---|---|---|
| Name: | Scholz | Butz | | |
| Date: | 23.08.91 | | | |
| Signature: | D. Scholz | | | |

## Table of Contents

## List of References

[1] Hoak, D.E., et al: USAF Stability and Control Datcom. Flight Control Division, Air Force Flight Dynamics Laboratory, WPAFB, Ohio, 1978.

[2] Roskam, J.: Airplane Design, Part VI. Roskam Aviation and Engineering Corporation, Rt4, Box 274, Ottawa, Kansas, 66067, USA, 1989.

[3] Backé, W.: Servo Steuerungen. Institut für hydraulische und pneumatische Antriebe und Steuerungen der RWTH Aachen, D-5100 Aachen.

[4] H. Schlichting, E. Truckenbrodt: Aerodynamik des Flugzeuges, Band 1. Springer, 1967.

**Actuator Installation**

For the actuator design it is important to know the actuator stroke and the effective lever arm as a function of surface deflection. These values can be calculated from data taken from installation drawings. A schematic is given in Fig. 1 with

| | | |
|---|---|---|
| a | : | distance between actuator pivot point and control surface pivot point |
| b | : | distance between actuator attachment points |
| $b_0$ | : | nominal distance between actuator attachment points at neutral position |
| c = r | : | measured lever arm length |
| $\alpha$ | : | action angle (angle between moment arm and actuator piston rod) |
| $\alpha_0$ | : | action angle with actuator in neutral position |
| $\beta$ | : | see Fig. 1 |
| $\beta_0$ | : | $\beta$ for control surface in neutral position |
| $\delta$ | : | control surface deflection |
| $r_{eff}$ | : | effective lever arm |
| x | : | piston rod travel in relation to actuator neutral position |
| s | : | actuator stroke |

The distance between actuator pivot point and control surface pivot point is a fixed value $a = a_0$. With equations as follows actuator stroke and effective lever arm can be calculated.

$$a = \sqrt{b_0^2 + c^2 - 2b_0c \cdot \cos\alpha_0} \tag{1}$$

$$\beta_0 = \arccos\left(\frac{b_0^2 - c^2 - a^2}{-2ac}\right) \tag{2}$$

$$\beta = \beta_0 + \delta \tag{3}$$

$$b = \sqrt{c^2 + a^2 - 2ac \cdot \cos\beta} \tag{4}$$

$$x = b - b_0 \tag{5}$$

$$r_{eff} = r \cdot \sin\alpha \tag{6}$$

Appendix A lists a FORTRAN program which uses Equations (1) to (~~5~~ 6) for writing and plotting values for stroke and effective lever arm. For plotting DISSPLA - subroutines are used, available on the Apollo workstation. As example the elevator is taken.

## Hinge Moments

Hinge moments are needed to compute actuator force levels so that the actuators can be properly sized. Normally nonlinear effects are encountered so that windtunnel data must be applied wherever possible. If for some reason windtunnel data is not available "Datcom" [1] or "Roskam: Airplane Design" [2] can be consulted.

Following [2], quick results can be obtained from a two dimensional approach, if assumptions as follows do apply:

- subsonic aircraft speed
- linear range of control
  (surface deflection < 20° ; angle of attack < 12°)
- the air flow over the surface is attached
- the airfoil is not cambered.

The hinge moment **HM** can be calculated from the hinge moment coefficient $\mathbf{c_h}$, the dynamic pressure $\mathbf{q} = \frac{1}{2}\,\rho\, v_{TAS}^2$ and the area of the control surface A

$$HM = c_h \cdot q \cdot A \cdot c \tag{7}$$

| | | |
|---|---|---|
| c | : | mean chord of control surface |
| $v_{TAS}$ | : | aircraft speed |
| $\rho$ | : | density of the air from Table 1. |

The hinge moment coefficient for an uncambered airfoil without tabs is calculated from

$$c_h = c_{h_\alpha} \cdot \alpha + c_{h_\delta} \cdot \delta \tag{8}$$

Depending on the application, the following substitutions must be made in Equation (~~10~~ 8):

| | | | |
|---|---|---|---|
| For a wing with aileron: | $\alpha = \alpha_w$ | ; $\delta = \delta_a$ | . |
| For a horizontal tail with elevator: | $\alpha = \alpha_h$ | ; $\delta = \delta_e$ | . |
| For a vertical tail with rudder: | $\alpha = \beta$ | ; $\delta = \delta_r$ | . |

The hinge moment derivative due to angle of attack $c_{h\alpha}$ is calculated from

$$c_{h_\alpha} = \left\{ \frac{c_{h_\alpha}'}{(c_{h_\alpha})_{theory}} \cdot (c_{h_\alpha})_{theory} + 2\,(c_{l_\alpha})_{theory} \cdot \left[ 1 - \frac{c_{l_\alpha}}{(c_{l_\alpha})_{theory}} \right] \cdot \left[ \tan\frac{\Phi''}{2} - \frac{t}{c} \right] \right\} \cdot \frac{(c_{h_\alpha})_{bal}}{c_{h_\alpha}''} \cdot \frac{1}{\sqrt{1 - M^2}} \tag{9}$$

with $\Phi''$ from Fig. 2 and the various factors from Fig. 3, 4, and 5.

The Reynolds Number **Re** (call $\mathbf{R_N}$ in Fig. 3) is

$$Re = \frac{v_{TAS} \cdot c}{\nu} \tag{10}$$

| | | |
|---|---|---|
| c | : | chord of wing, horizontal tail, or vertical tail |
| $\nu$ | : | kinematic viscosity from Table 1. |

The Mach Number **M** is

$$M = \frac{v_{TAS}}{a} \tag{11}$$

Where **a** is the speed of sound from Table 1.

The hinge moment derivative due to control surface deflection $c_{h\delta}$ is calculated from

$$c_{h_\delta} = \left\{ \frac{c_{h_\delta}'}{(c_{h_\delta})_{theory}} \cdot (c_{h_\delta})_{theory} + 2\,(c_{l_\delta})_{theory} \cdot \left[ 1 - \frac{c_{l_\delta}}{(c_{l_\delta})_{theory}} \right] \cdot \left[ \tan\frac{\Phi''}{2} - \frac{t}{c} \right] \right\} \cdot \frac{(c_{h_\delta})_{bal}}{c_{h_\delta}''} \cdot \frac{1}{\sqrt{1 - M^2}} \tag{12}$$

with the various additional factors from Fig. 6, 7, and 8.

Hinge moments have to be calculated for the various flight conditions in order to find the maximum value. For these various flight conditions the angles of attack and the required control surface deflection rates have to be determined before the hinge moment can be calculated from Equation (~~9~~ 7).

The maximum elevator hinge moment has to be found among elevator hinge moments calculated for various load factors and altitudes at maximum operating speed ($V_{Mo}$) or maximum operating Mach Number ($M_{Mo}$).

The maximum aileron hinge moment can be calculated for an altitude of 0 ft, $v_{Mo}$ / $M_{Mo}$ - condition, and maximum aileron deflection.

Rudder hinge moments need to be calculated for various $v_{Mo}$ / $M_{Mo}$ - conditions and various altitudes and maximum rudder deflection. Maximum rudder deflection is a function of rudder travel limitation which again depends on aircraft speed and permissible structural loads.

For spoiler hinge moments the approach shown above does <u>not</u> work because spoilers are attached on the wing instead of forming the trailing edge. Spoiler hinge moments need to be calculated for retracted and extended surface. Hinge moments for the retracted position are calculated with flaps extended. Hinge moments for extended spoilers are calculated for the aircraft at emergency decent speed and a low altitude. The spoiler hinge moment is

$$HM = c_N \cdot A \cdot q \cdot \frac{x}{c} \cdot c \tag{13}$$

with

| | | |
|---|---|---|
| $c_N$ | : | normal force coefficient, positive downwards, acts normal to reference plane. |
| A | : | spoiler area |
| x/c | : | center of pressure as a fraction of chord of element measured form leading edge of element |
| c | : | chord of the spoiler |

$c_N$ has to be obtained from aerodynamic wing data.

<u>**Actuator Design**</u>

The actuator stroke s can already be determined from the installation data as shown above. The actuator piston area **A** is

$$A = \frac{SF \cdot HM_{max}}{r_{eff} \cdot \Delta p} \tag{14}$$

with

| | | |
|---|---|---|
| SF | : | safety factor |
| $H.M._{max}$ | : | maximum hinge moment (all possible flight conditions considered) |
| $r_{eff}$ | : | effective lever arm |
| $\Delta p$ | : | pressure difference at actuator piston |

The pressure difference at the actuator piston has to be selected carefully taking into account the delivery pressure of the hydraulic pump at realistic flow conditions together with the pressure losses in the valve and those in the supply and return lines.

For an actuator with balanced piston, the diameter of the piston rod **d** and the diameter of the piston **D** can be calculated from the ratio **D/d**. For a preliminary design standard values for **D/d** can be selected. The piston rod diameter is

$$d = \sqrt{\frac{4/\pi \cdot A}{(D/d)^2 - 1}} \tag{15}$$

The instantaneous hydraulic flow without leakage of the actuator is

$$Q = \omega \cdot \frac{2\pi}{360} \cdot r_{eff} \cdot A \tag{16}$$

where $\omega$ is the deflection rate of the surface in °/s . An average hydraulic flow can be calculated from

$$Q = \omega \cdot \frac{s}{\delta_{max} - \delta_{min}} \cdot A \tag{17}$$

Stall force $F_{stall}$ and stall hinge moment $HM_{stall}$ are calculated from the pressure difference selected for stall conditions $\Delta p_{stall}$ :

$$F_{stall} = \Delta p_{stall} \cdot A \tag{18}$$

$$HM_{stall} = F_{stall} \cdot r_{eff} \tag{19}$$

The damping hinge moment $HM_{damp}$ is calculated from the damping coefficient $c_{damp}$

$$HM_{damp} = c_{damp} \cdot \omega^2 \tag{20}$$

The damping hinge moment from the actuator in damping mode requires a certain pressure difference $\Delta p_{damp}$ in the parallel and active actuator. The pressure drop in the servo valve $\Delta p_{servo}$ of the active actuator is the difference to the nominal operating pressure $\Delta p_{nominal,operating}$ :

$$\Delta p_{servo} = \Delta p_{nominal,operating} - \Delta p_{damp} \tag{21}$$

with

$$\Delta p_{damp} = \frac{c_{damp} \cdot \omega^2}{r_{eff} \cdot A} \tag{22}$$

Pressure drop in the servo valve and flow at maximum deflection rate $Q_{max}$ of the surface (aircraft on ground) are the design values for the opening area of the ports in the servo valve $A_{opening}$ :

$$A_{opening} = \frac{Q_{max}}{c_d \cdot \sqrt{2 / \rho \cdot \Delta p_{servo} / 2}} \tag{23}$$

with

$c_d$ : loss coefficient
$\rho$ : density of hydraulic fluid

For a preliminary design, the valve spool stroke s and the valve spool diameter **d** can be calculated from standard values **d/s**, keeping in mind that $A_{opening} = \pi \cdot d \cdot s$ hence

$$s = \sqrt{\frac{A_{opening}}{\pi \cdot d/s}} \tag{24}$$

If the pressure difference at the actuator ports $\Delta p_+$ is higher than required for the present hinge moment, the surface will move with a certain deflection rate $\omega$. Only the damping force of the parallel actuator (requiring a pressure difference $\Delta p_{damp}$ in the active actuator) and the pressure drop in the servo valve $\Delta p_{servo}$ are considered in Equation (~~29~~ 25). With

$$\Delta p_+ = \Delta p_{damp} + \Delta p_{servo} \tag{25}$$

$\Delta p_{damp}$ from Equation (~~24~~ 22) and $\Delta p_{servo}$ determined from Equations (~~28~~ 21)

$$Q = A_{opening} \cdot c_d \cdot \sqrt{\frac{2}{\rho} \cdot \frac{\Delta p_{servo}}{2}} \tag{26}$$

$$Q = \omega \cdot \frac{s}{\delta_{max} - \delta_{min}} \cdot A$$

allows the calculation of the deflection rate

$$\omega = \sqrt{\frac{\Delta p_+}{\frac{c_{damp}}{r_{eff} \cdot A_{actuator}} + \left(\frac{A_{actuator}}{A_{opening}} \cdot \frac{s}{\delta_{max} - \delta_{min}} \cdot \frac{1}{c_d}\right)^2 \cdot \rho}} \tag{27}$$

A good understanding of the pressure in a servo valve at various over- and underlap conditions can be obtained by plotting the pressure versus valve spool stroke. Summing the pressure in actuator chamber A and actuator chamber B shows overlap or underlap conditions as presented in Fig. 9. It is assumed that there is no flow into or out of the actuator chambers. With nomenclature form Fig. 9, the pressure in the actuator chambers A and B can be calculated as shown in the following derivation.

$$Q_1 \sim s_1 \cdot \sqrt{p_0 - p_A} \tag{28}$$

$$Q_2 \sim s_2 \cdot \sqrt{p_0 - p_B} \tag{29}$$

$$Q_3 \sim s_3 \cdot \sqrt{p_A} \tag{30}$$

$$Q_4 \sim s_4 \cdot \sqrt{p_B} \tag{31}$$

$$Q_1 = Q_3 \tag{32}$$

$$Q_2 = Q_4 \tag{33}$$

$$s_1 = y_{01E} + y \tag{34}$$

$$s_2 = y_{02E} - y \tag{35}$$

$$s_3 = y_{01A} - y \tag{36}$$

$$s_4 = y_{02A} + y \tag{37}$$

$$p_A = p_0 \cdot \frac{1}{1 + \left(\frac{s_3}{s_1}\right)^2} \tag{38}$$

$$p_B = p_0 \cdot \frac{1}{1 + \left(\frac{s_4}{s_2}\right)^2} \tag{39}$$

A FORTRAN program for plotting the pressure diagrams is given in Appendix B. For plotting this program uses DISSPLA - subroutines available on the Apollo workstation.

# Tables and Figures

*Luftdruck p, Luftdichte ϱ, Temperatur T, Schallgeschwindigkeit a und kinematische Zähigkeit ν in Abhängigkeit von der Höhe z für die US Standard Atmosphäre*

| z [km] | $T/T_0$ | $p/p_0$ | $\varrho/\varrho_0$ | $a/a_0$ | $\nu/\nu_0$ |
|---|---|---|---|---|---|
| 0 | 1,0 | 1,0 | 1,0 | 1,0 | 1,0 |
| 2 | 0,9549 | $7{,}846 \cdot 10^{-1}$ | $8{,}217 \cdot 10^{-1}$ | 0,9772 | 1,174 |
| 4 | 0,9097 | $6{,}085 \cdot 10^{-1}$ | $6{,}688 \cdot 10^{-1}$ | 0,9538 | 1,388 |
| 6 | 0,8647 | $4{,}660 \cdot 10^{-1}$ | $5{,}389 \cdot 10^{-1}$ | 0,9299 | 1,654 |
| 8 | 0,8197 | $3{,}518 \cdot 10^{-1}$ | $4{,}292 \cdot 10^{-1}$ | 0,9054 | 1,988 |
| 10 | 0,7747 | $2{,}615 \cdot 10^{-1}$ | $3{,}376 \cdot 10^{-1}$ | 0,8802 | 2,413 |
| 11,019 | 0,7519 | $2{,}234 \cdot 10^{-1}$ | $2{,}971 \cdot 10^{-1}$ | 0,8671 | 2,674 |
| 12 | 0,7519 | $1{,}915 \cdot 10^{-1}$ | $2{,}546 \cdot 10^{-1}$ | 0,8671 | 3,120 |
| 14 | 0,7519 | $1{,}399 \cdot 10^{-1}$ | $1{,}860 \cdot 10^{-1}$ | 0,8671 | 4,271 |
| 16 | 0,7519 | $1{,}022 \cdot 10^{-1}$ | $1{,}359 \cdot 10^{-1}$ | 0,8671 | 5,846 |
| 18 | 0,7519 | $7{,}466 \cdot 10^{-2}$ | $9{,}930 \cdot 10^{-2}$ | 0,8671 | 8,000 |
| 20 | 0,7519 | $5{,}457 \cdot 10^{-2}$ | $7{,}258 \cdot 10^{-2}$ | 0,8671 | $1{,}095 \cdot 10^{1}$ |
| 20,063 | 0,7519 | $5{,}403 \cdot 10^{-2}$ | $7{,}186 \cdot 10^{-2}$ | 0,8671 | $1{,}106 \cdot 10^{1}$ |
| 25 | 0,7689 | $2{,}516 \cdot 10^{-2}$ | $3{,}272 \cdot 10^{-2}$ | 0,8769 | $2{,}474 \cdot 10^{1}$ |
| 30 | 0,7861 | $1{,}181 \cdot 10^{-2}$ | $1{,}503 \cdot 10^{-2}$ | 0,8866 | $5{,}486 \cdot 10^{1}$ |
| 32,162 | 0,7935 | $8{,}567 \cdot 10^{-3}$ | $1{,}080 \cdot 10^{-2}$ | 0,8908 | $7{,}696 \cdot 10^{1}$ |
| 35 | 0,8208 | $5{,}671 \cdot 10^{-3}$ | $6{,}909 \cdot 10^{-3}$ | 0,9060 | $1{,}236 \cdot 10^{2}$ |
| 40 | 0,8688 | $2{,}834 \cdot 10^{-3}$ | $3{,}262 \cdot 10^{-3}$ | 0,9321 | $2{,}743 \cdot 10^{2}$ |
| 45 | 0,9168 | $1{,}472 \cdot 10^{-3}$ | $1{,}605 \cdot 10^{-3}$ | 0,9575 | $5{,}819 \cdot 10^{2}$ |
| 47,350 | 0,9393 | $1{,}095 \cdot 10^{-3}$ | $1{,}165 \cdot 10^{-3}$ | 0,9692 | $8{,}170 \cdot 10^{2}$ |
| 50 | 0,9393 | $7{,}874 \cdot 10^{-4}$ | $8{,}383 \cdot 10^{-4}$ | 0,9692 | $1{,}136 \cdot 10^{3}$ |
| 52,429 | 0,9393 | $5{,}823 \cdot 10^{-4}$ | $6{,}199 \cdot 10^{-4}$ | 0,9692 | $1{,}536 \cdot 10^{3}$ |
| 55 | 0,9218 | $4{,}219 \cdot 10^{-4}$ | $4{,}578 \cdot 10^{-4}$ | 0,9601 | $2{,}049 \cdot 10^{3}$ |
| 60 | 0,8876 | $2{,}217 \cdot 10^{-4}$ | $2{,}497 \cdot 10^{-4}$ | 0,9421 | $3{,}645 \cdot 10^{3}$ |
| 61,591 | 0,8768 | $1{,}797 \cdot 10^{-4}$ | $2{,}050 \cdot 10^{-4}$ | 0,9364 | $4{,}397 \cdot 10^{3}$ |
| 65 | 0,8305 | $1{,}130 \cdot 10^{-4}$ | $1{,}360 \cdot 10^{-4}$ | 0,9113 | $6{,}340 \cdot 10^{3}$ |
| 70 | 0,7625 | $5{,}448 \cdot 10^{-5}$ | $7{,}146 \cdot 10^{-5}$ | 0,8732 | $1{,}125 \cdot 10^{4}$ |
| 75 | 0,6946 | $2{,}458 \cdot 10^{-5}$ | $3{,}538 \cdot 10^{-5}$ | 0,8334 | $2{,}100 \cdot 10^{4}$ |
| 79,994 | 0,6269 | $1{,}024 \cdot 10^{-5}$ | $1{,}634 \cdot 10^{-5}$ | 0,7918 | $4{,}161 \cdot 10^{4}$ |
| 80 | 0,6269 | $1{,}023 \cdot 10^{-5}$ | $1{,}632 \cdot 10^{-5}$ | 0,7918 | $4{,}166 \cdot 10^{4}$ |
| 85 | 0,6269 | $4{,}071 \cdot 10^{-6}$ | $6{,}494 \cdot 10^{-6}$ | 0,7918 | $1{,}047 \cdot 10^{5}$ |
| 90 | 0,6269 | $1{,}622 \cdot 10^{-6}$ | $2{,}588 \cdot 10^{-6}$ | 0,7918 | $2{,}627 \cdot 10^{5}$ |

$g_0 = 9{,}8067\ \mathrm{m/s^2}$, $T_0 = 288{,}15\ °\mathrm{K}$,

$p_0 = 10332\ \mathrm{kp/m^2}$, $t_0 = 15\ °\mathrm{C}$,

$\varrho_0 = 1{,}2250\ \mathrm{kg/m^3}$, $a_0 = 340{,}29\ \mathrm{m/s}$,

$\varrho_0 = 0{,}1249\ \mathrm{kp\ s^2/m^4}$, $\nu_0 = 1{,}4607 \cdot 10^{-5}\ \mathrm{m^2/s}$,

$$(dT/dH)_0 = -6{,}5\ \mathrm{grd/km}.$$

Table 1 from [4]







Fig. 1 Schematic with values for actuator installation




Fig. 2 Definition of Trailing Edge Angles [2]




Fig. 3 **Effect of Airfoil Thickness and Trailing Edge Angle on Lift Curve Slope**

from [2]




Fig. 4 **Two-Dimensional Control Surface Hingemoment Derivative due to Angle of Attack**

from [2]

$c_b/c_f$ IS CALLED THE OVERHANG




Effect of Nose Shape and Balance on the Two-Dimensional Hingemoment Derivative due to Angle of Attack




Fig. 5

Nose Shape Examples for 35 Percent Balance

from [2]



1.0
.8
.6
.4
0
.1
.2
.3
.4
.5
$c_{\ell_\delta} / (c_{\ell_\delta})_{theory}$
1.00
.98
.96
.94
.92
.90
.88
.86
.84
.82
.80
.78
.76
.74
.72
.70
$c_{\ell_\alpha} / (c_{\ell_\alpha})_{theory}$
$c_f/c$




t/c
.15
.12
.10
.08
.06
.04
.02
0
6
5
4
3
2
0
.1
.2
.3
.4
.5
$(c_{\ell_\delta})_{theory}$ (per rad)
$c_f/c$
c
$c_f$
t
$\delta_f$


Fig. 6 from [2]




Fig. 7 Two-Dimensional Control Surface Hingemoment Derivative due to Control Surface Deflection

from [2]




Fig. 8 — Effect of Nose Shape and Balance on the Two-Dimensional Control Surface Hingemoment Derivative due to Control Surface Deflection

from [2]




Fig. 9 Pressure in actuator chamber A and B at various positions of the valve spool and over- and underlap conditions [3].

```
C   ************
C   * ELEVATOR *
C   ************

      REAL a,b,c,b0,al,be,dl,al0,be0
      REAL xdl,yb,ystroke,yarm
      INTEGER i,m
      DIMENSION b(20),al(20),be(20),dl(20)
      DIMENSION xdl(101),yb(101),ystroke(101),yarm(101)

      DATA c,b0 /76.0,390.0/
      DATA dl /-18.0,-15.0,-10.55,-10.0,-5.0,0.0,5.0,10.0,15.0,20.0,
     1          25.0,30.0,33.0,7*0.0/
      DATA al0 /78.0/
      m   = 13

      DO 10 i = 1,m
        dl(i) = RAD(dl(i))
   10 CONTINUE
      al0 = RAD(al0)

      a   = SQRT(b0**2 + c**2 - 2.0*b0*c*COS(al0))
      be0 = ACOS((b0**2 - c**2 - a**2)/(-2.0*a*c))

      WRITE(6,100)
  100 FORMAT(///1H ,'ELEVATOR'/
     1          1H ,'========'//
     2          1H ,'SURFACE           ','STROKE             ',
     3              'ACTION ANGLE      ','MOMENT ARM         '/
     4          1H ,'DEFLECTION [DEG] ','[MM]               ',
     5              '[DEG]             ','[MM]               '/
     6          1H ,64(1h-))

      DO 20 i = 1,m
        be(i)   = be0 + dl(i)
        b(i)    = SQRT(c**2 + a**2 - 2*a*c*COS(be(i)))
        al(i)   = ACOS((a**2 - b(i)**2 - c**2)/(-2.0*b(i)*c))
      WRITE(6,200) DEG(dl(i)), b(i)-b(1), DEG(al(i)), c*SIN(al(i))
  200 FORMAT(1H ,F10.1,3F17.1)
   20 CONTINUE
      WRITE(6,300)
  300 FORMAT(///)

      xorig = -50.0
      xmax =   50.0
      xstep = (xmax - xorig)/100.0
      xdlorig = -18.0
      xdlmax  =  33.0
      xdlstep = (xdlmax - xdlorig)/100.0
      yorig = 0.0
      ymax = 100.0
      ystep = (ymax - yorig)/100.0

      DO 30 i = 1,101
      xdl(i)      = xdlorig + xdlstep*(i - 1)
      ybe         = be0 + RAD(xdl(i))
      yb(i)       = SQRT(c**2 + a**2 - 2*a*c*COS(ybe))
      ystroke(i)  = yb(i) - yb(1)
      yarm(i)     = c*SIN(ACOS((a**2 - yb(i)**2 - c**2)/(-2.0*yb(i)*c)))
   30 CONTINUE

      CALL YNAME('stroke, momentarm [mm]',22)
      CALL XNAME('surface deflection [deg]',24)

  400 CALL GRAF(xorig,xstep,xmax,yorig,ystep,ymax)
      CALL MARKER('S')
      CALL CURVE(xdl,ystroke,101,+10)
      CALL MARKER('A')
      CALL CURVE(xdl,yarm,101,+10)
      CALL MESSAG('ELEVATOR',8,-40.0,70.0)
      CALL MESSAG('S = stroke [mm]',15,-40.0,60.0)
      CALL MESSAG('A = moment arm [mm]',19,-40.0,50.0)
      CALL KEY_STROKE(IFLAG)
      IF(IFLAG .EQ. 1) GOTO 400
      CALL PLTEND(IERROR)

      END


      FUNCTION rad(al)
```

```
pi = 3.14159265
rad = al/180.0*pi
RETURN
END

FUNCTION deg(al)
pi = 3.14159265
deg = al*180.0/pi
RETURN
END
```

# Appendix B

```
C*********************************************************************
C
C     Programm zur Berechnung der Druecke am Ausgang
C     eines Servoventils
C
C*********************************************************************
C
C ***  MNPTS: Anzahl der zu berechnenden Punkte fuer den Plot
C
      PARAMETER( MNPTS =1001 )
      DIMENSION Y(MNPTS), PA(MNPTS), PB(MNPTS), PS(MNPTS)
C
C ***  Alle Laengen in mm
C
C      YMIN : Stellung des Kolbens des Servoventils am
C             linken Anschlag
C      YMAX : Stellung des Kolbens des Servoventils am
C             rechten Anschlag
C      Y01A : Ueberdeckung aussen links
C      Y01E : Ueberdeckung innen links
C      Y02E : Ueberdeckung innen rechts
C      Y02A : Ueberdeckung aussen rechts
C
C      SPALT: Mass des verbleibenden (geringen) Spieles
C             des Servoventil-Kolbens bei Nullueberdeckung
C             !!! SPALT muss groesser als Null sein !!!
C
      DATA YMIN,YMAX / - 0.7, 0.7/
      DATA Y01A,Y01E,Y02E,Y02A /0.23240, 0.23240, 0.2324, 00.23240 /
      DATA SPALT /0.002 /
C
C *** Berechnet fuer p0 = 1.0
C
      NPTS  = MNPTS
      YSTEP = (YMAX - YMIN)/(NPTS - 1)
C
      DO 10 I=1,NPTS
         Y(I) = YMIN + YSTEP*(I-1)
         S1H  = 0.5*(Y01E + Y(I) + ABS(Y01E + Y(I)))
         S2H  = 0.5*(Y02E - Y(I) + ABS(Y02E - Y(I)))
         S3H  = 0.5*(Y01A - Y(I) + ABS(Y01A - Y(I)))
         S4H  = 0.5*(Y02A + Y(I) + ABS(Y02A + Y(I)))
         S1   = SQRT(S1H**2 + SPALT**2)
         S2   = SQRT(S2H**2 + SPALT**2)
         S3   = SQRT(S3H**2 + SPALT**2)
         S4   = SQRT(S4H**2 + SPALT**2)
         PA(I)= 1./(1. + (S3/S1)**2)
         PB(I)= 1./(1. + (S4/S2)**2)
         PS(I)= PA(I) + PB(I)

   10 CONTINUE
C
C *** Ausgabe spezieller Werte
C
      WRITE(6,200) Y(667),PA(667),PB(667),PS(667)
  200 FORMAT(//1H ,'Ausgabe der Werte fuer 33% Ventiloeffnung'//
     1        1H ,'Y  = ',F8.5/1H ,'PA = ',F8.5/1H ,'PB = ',F8.5/
     2        1H ,'PS = ',F8.5//)
C
C *** Plotten der Ergebnisse
C
      XORIG = -0.7
      XSTEP =  0.1
      XMAX  =  0.7
C
      YORIG =-0.1
      YSTEP = 0.1
      YMAX  = 1.5
C
      CALL XNAME('Y - Servoventil-Stellung [mm]',29)
      CALL YNAME('p/p0 - Druckverhaeltnis',23)
C
  100 CALL GRAF(XORIG,XSTEP,XMAX,YORIG,YSTEP,YMAX)
C
      CALL MARKER('A')
      CALL CURVE(Y,PA,NPTS,+100)
C
      CALL MARKER('B')
      CALL CURVE(Y,PB,NPTS,+100)
C
```

```
      CALL MARKER('S')
      CALL CURVE(Y,PS,NPTS,+200)
C
      CALL MESSAG('LAT Seitenruder/ABEX 410 modifiziert',36,-0.6,1.45)
      CALL MESSAG('Ventil:   SYMMETRISCH;  SPALT=0.002',35,-0.6,1.25)
      CALL MESSAG
     1    ('Y= +/- 0.7  Y01A= 0.232 Y01E= 0.232 Y02E= 0.232 Y02A= 0.232'
     2                                                 ,59, -0.6,1.05)
C
      CALL KEY_STROKE(IFLAG)
      IF ( IFLAG .EQ. 1)   GO TO 100
      CALL PLTEND(IERROR)
C
      STOP
      END
```




Ausgabe vom "Programm zur Berechnung der Drücke am Ausgang eines Servoventils"